Travaux dirigés avec SAGE (partie II)

Math 2 — Année 2010-2011

Sommaire

Conseil : pour plus de clarté, faites chaque exercice dans une feuille de calcul séparée.

1 Calcul d'intégrales

1.1 Primitives

On utilise la commande `integrate`. Entrez par exemple l'expression $y = \frac{1}{2+3\cos x}$:

```
var('x')
y = 1/(2+3*cos(x)) ; y
```

et calculez l'expression z d'une primitive de y :

```
z = integrate(y,x) ; z ; show(z)
```

Pour vérification, calculez la dérivée de z par rapport à x :

```
dz = diff(z,x) ; dz ; show(dz)
```

Il n'est pas clair que $z' = y$, à moins de forcer la simplification complète de cette dernière expression :

```
dz.simplify_full()
```

1.2 Intégrales définies

Calculez $I = \int_0^{\pi/2} \frac{1}{2+3\cos(x)}\,dx$:

```
I = integrate(y,x,0,pi/2) ; I
```

On peut aussi calculer la valeur approchée d'une intégrale. Entrez la commande :

```
numerical_integral(y,0,pi/2)
```

qui donne un couple de valeurs : (*approximation*, *majoration d'erreur*). Comparez avec :

```
I.n(digits=16)
```

1.3 Exercices

1. Calculer les intégrales :

$$A = \int_0^1 \frac{x}{(x^2+1)\sqrt{1-x^4}}\,dx \quad , \quad B = \int_0^1 \frac{\sqrt{1+x}}{1+\sqrt[3]{1+x}}\,dx$$

2. On considère l'intégrale :

$$C = \int_1^2 \frac{1}{x\,\sqrt{1+x^{16}+x^{32}}}\,dx$$

   (a) Vérifier que le calcul exact de cette intégrale échoue.

   (b) Donner une approximation numérique de C.

   (c) On se propose de calculer C en effectuant le changement de variable :

$$x = t^{-1/16} \tag{1}$$

   En remplaçant x par $t^{-1/16}$ dans l'expression à intégrer et en calculant la dérivée de x par rapport à t avec (1), former une intégrale en t égale à C. En déduire la valeur exacte de C.

   (d) Vérifier le résultat grâce à la question (b).

2 Intégrales doubles

2.1 Principe

SAGE n'a pas de commande spécifique pour calculer les intégrales multiples. Nous calculerons ces dernières par emboîtement d'intégrales simples. On pourra s'aider d'une visualisation graphique du domaine d'intégration.

Soit $\mathcal{C}$ la courbe plane d'équation $x^3 + x^2 - y^2 = 0$. Commencez par tracer $\mathcal{C}$:

```
var('x,y')
equ = x^3+x^2-y^2==0
implicit_plot(equ,(x,-2,2),(y,-2,2),aspect_ratio=1)
```

Déterminez les points d'intersection de $\mathcal{C}$ avec l'axe des x :

```
solve([equ,y==0],x,y)
```

On trouve les points $(-1, 0)$ et $(0, 0)$.

Pour un x donné, calculez les valeurs de y pour lesquelles $(x, y) \in \mathcal{C}$:

```
solve(equ,y)
```

d'où les deux valeurs :

```
a = x*sqrt(x+1) ; b = -a
```

Soit Δ le domaine fermé borné délimité par la boucle de $\mathcal{C}$. D'après ce qui précède, si f est une fonction de (x, y), on a :

$$\iint_\Delta f(x,y)\,dx\,dy = \int_{-1}^{0} \left(\int_{x\sqrt{x+1}}^{-x\sqrt{x+1}} f(x,y)\,dy \right) dx \tag{2}$$

2.2 Applications

Calculez l'aire de Δ en appliquant la formule (2) :

```
A = integrate(integrate(1,y,a,b),x,-1,0) ; A
```

SAGE n'y parvient pas car il ignore que $\sqrt{x+1}$ est réel. Aidez-le en entrant la contrainte :

```
assume(x+1>=0)     # signifie qu'on suppose x+1 positif
```

et cette fois le calcul aboutit :

```
A = integrate(integrate(1,y,a,b),x,-1,0) ; A
```

A tout instant, on peut voir les contraintes déjà effectuées avec la commande :

```
assumptions()
```

Astuce.— Une façon de contourner le problème est de passer par une primitive, comme ceci :

```
u = integrate(1,y) ; v = u(y=b)-u(y=a)
A = integrate(v,x,-1,0) ; A
```

Calculez les coordonnées du centre de masse de Δ (en supposant la densité surfacique égale à 1) :

```
xG = 1/A * integrate(integrate(x,y,a,b),x,-1,0)
yG = 1/A * integrate(integrate(y,y,a,b),x,-1,0)
xG ; yG
```

On suppose maintenant que la densité surfacique est $\sigma(x, y) = x^2 + y\,(y + 1)$. Calculez la masse de Δ :

```
sigma(x,y) = x^2+y*(y+1)
M = integrate(integrate(sigma(x,y),y,a,b),x,-1,0) ; M
```

et les coordonnées de son centre de masse :

```
xG = 1/M * integrate(integrate(x*sigma(x,y),y,a,b),x,-1,0)
yG = 1/M * integrate(integrate(y*sigma(x,y),y,a,b),x,-1,0)
xG ; yG
```

Pour la suite, levez toutes les contraintes avec la commande :

```
forget()
```

2.3 Exercice

On se propose de calculer l'intégrale double :

$$I = \iint_D \frac{y}{\sqrt{x^2 + y^2}}\,dx\,dy \tag{3}$$

où D est le domaine du demi-plan $y \geqslant 0$ délimité par les quatre paraboles d'équations :

$$\mathcal{P}_1 : y^2 - 4x - 4 = 0\ ,\ \mathcal{P}_2 : y^2 - 2x - 1 = 0\ ,\ \mathcal{P}_3 : y^2 + 6x - 9 = 0\ ,\ \mathcal{P}_4 : y^2 + 4x - 4 = 0.$$

1. Entrer les équations des paraboles en complétant la cellule :

```
var('x,y')
eq1 = y^2-4*x-4 == 0
eq2 = ...
```

2. Tracer sur un même dessin les quatre paraboles pour $-1 \leqslant x \leqslant 2$ et $0 \leqslant y \leqslant 3$.
3. On envisage le changement de variables défini par :

$$x = u^2 - v^2, y = 2uv \quad (u > 0, v > 0) \tag{4}$$

Entrer :

```
var('u,v')
eq1(x=u^2-v^2,y=2*u*v).factor()
```

En déduire que la parabole $\mathcal{P}_1$ est transformée en la droite d'équation $v = 1$. Vérifiez de même que les paraboles $\mathcal{P}_2, P_3, P_4$ sont transformées en des droites que l'on précisera.

4. Calculer le déterminant jacobien du changement de variable (4) avec la commande :

```
x = u^2-v^2 ; y = 2*u*v
dJ = det(jacobian((x,y),(u,v))) ; dJ
```

5. Déduire des questions précédentes que $I = \iint_\Delta 8\,uv\,du\,dv$, où Δ est le domaine $1 \leqslant u \leqslant \sqrt{3/2}$, $1/\sqrt{2} \leqslant v \leqslant 1$.
6. Conclure.

3 Intégrales triples

3.1 Principe

Soit T le tétraèdre défini par $x \geqslant 0, y \geqslant 0, z \geqslant 0$ et $x + y + z \leqslant a$ (où a est > 0), supposé homogène de masse volumique égale à 1. Dessinez la face supérieure de T en prenant par exemple $a = 1$:

```
var('x,y,z')
implicit_plot3d(x+y+z==1,(x,0,1),(y,0,1),(z,0,1))
```

Pour calculer le moment d'inertie m de T par rapport à l'origine, qui est donné par l'intégrale :

$$m = \iiint_T (x^2 + y^2 + z^2)\,dx\,dy\,dz = \int_{x=0}^{a} \int_{y=0}^{a-x} \int_{z=0}^{a-x-y} (x^2 + y^2 + z^2)\,dz\,dy\,dx$$

il suffit d'entrer la commande :

```
var('a')
u = integrate(x^2+y^2+z^2,z,0,a-x-y)
v = integrate(u,y,0,a-x)
m = integrate(v,x,0,a)
m
```

qui donne $m = \frac{a^5}{20}$.

3.2 Exercices

1. Calculer la position du centre de gravité d'une demi-boule homogène B définie par $x^2+y^2+z^2 \leqslant 1$ et $z \geq 0$. *Indication* : utiliser des coordonnées sphériques, pour lesquelles :

$$\iiint_B f(x, y, z)\,dx\,dy\,dz = \int_{\theta=0}^{2\pi} \int_{\varphi=0}^{\frac{\pi}{2}} \int_{\rho=0}^{1} f(\rho\cos\theta\cos\varphi, \rho\sin\theta\cos\varphi, \rho\sin\varphi)\,\rho^2\cos\varphi\,d\rho\,d\varphi\,d\theta$$

2. Une boule centrée à l'origine est percée d'un trou cylindrique vertical d'axe Oz. Le solide S qui en résulte est de hauteur h. Montrer que le volume de S est égal à $\frac{1}{6}\pi h^3$, donc indépendant du rayon R de la boule. Calculer le moment d'inertie de S par rapport à l'axe des z en fonction de R et h (on suppose la densité volumique constante égale à 1). *Indication* : utiliser des coordonnées cylindriques, pour lesquelles :

$$\iiint_S f(x,y,z)\,dx\,dy\,dz = \int_{\theta=0}^{2\pi}\int_{\rho=\sqrt{R^2-h^2/4}}^{R}\int_{z=-\sqrt{R^2-\rho^2}}^{\sqrt{R^2-\rho^2}} f(\rho\cos\theta,\rho\sin\theta,z)\,\rho\,dz\,d\rho\,d\theta$$

ainsi que l'astuce du §2.2.



4 Intégrales curvilignes

4.1 Courbe paramétrée plane

Nous prendrons comme support la courbe $\mathcal{C}$ du §2.1. Trouvons une paramétrisation de $\mathcal{C}$ en cherchant les points d'intersection de $\mathcal{C}$ avec une droite « tournante » d'équation $y = tx$ (t paramètre réel). Pour cela, entrez la commande :

```
var('x,y,t')
equ = x^3+x^2-y^2==0
solve([equ,y==t*x],[x,y])
```

d'où l'on tire la paramétrisation :

```
x(t) = t^2-1 ; y(t) = t^3-t
```

Pour vérification, tracez cette courbe paramétrée :

```
parametric_plot((x(t),y(t)),(t,-3/2,3/2),aspect_ratio=1)
```

On note $\mathcal{C}_a^b$ l'arc de la courbe $\mathcal{C}$ correspondant à t variant de a à b. La longueur de l'arc $\mathcal{C}_{-1}^1$ (boucle de $\mathcal{C}$) est donnée par l'intégrale $\int_{-1}^1 \sqrt{x'(t)^2+y'(t)^2}\,dt$:

```
ds = sqrt(diff(x(t),t)^2+diff(y(t),t)^2)
numerical_integral(ds,-1,1)
```

4.2 Intégrale le long d'un arc de courbe

Définissons deux fonctions $\vec{F_i} : (u,v) \mapsto (P_i(u,v), Q_i(u,v))$ pour $i = 1, 2$:

```
var('u,v')
P1(u,v) = 2*u*v-u^2 ; Q1(u,v) = u+v^2
P2(u,v) = 6*u*v^2-v^3 ; Q2(u,v) = 6*u^2*v-3*u*v^2
```

Calculez :

```
diff(P1(u,v),v)-diff(Q1(u,v),u) ; diff(P2(u,v),v)-diff(Q2(u,v),u)
```

Qu'en concluez-vous ? Vérifiez que $\vec{F_2} = \vec{\nabla} f$ avec $f(u,v) = 3u^2v^2 - uv^3$:

```
f(u,v) = 3*u^2*v^2-u*v^3
f(u,v).gradient()
```

Calculez l'intégrale de $\vec{F_1}$ le long de l'arc $\mathcal{C}_1^2$ allant du point $O = (x(1), y(1)) = (0,0)$ au point $M = (x(2), y(2)) = (3,6)$, donnée par l'intégrale $\int_1^2 \big(P_1(x(t), y(t))\, x'(t) + Q_1(x(t), y(t))\, y'(t)\big)\, dt$:

```
integrate(P1(x(t),y(t))*diff(x(t),t)+Q1(x(t),y(t))*diff(y(t),t),t,1,2)
```

Que se passe-t-il si on prend un autre chemin de O à M (par exemple un segment de droite) ?

Faites le même travail avec $\vec{F_2}$ et comparez le résultat avec $f(3,6) - f(0,0)$. Que remarquez-vous ?

Calculez de même les intégrales de $\vec{F_1}$ et $\vec{F_2}$ le long de l'arc fermé $\mathcal{C}_{-1}^1$. Le résultat était-il prévisible ?

4.3 Formule de Green-Riemann

En procédant comme au §2.2, calculez l'intégrale double :

$$\iint_\Delta \left(\frac{\partial Q_1(x,y)}{\partial x} - \frac{\partial P_1(x,y)}{\partial y} \right) dx\, dy$$

Pourquoi retrouve-t-on la valeur de l'intégrale de $\vec{F_1}$ le long de $\mathcal{C}_{-1}^1$?

Calculez enfin l'intégrale :

$$\frac{1}{2} \int_{-1}^{1} \big(x(t)y'(t) - y(t)x'(t)\big)\, dt$$

Pourquoi retrouve-t-on l'aire de Δ calculée au §2.2 ?

4.4 Exercice

Soit $\mathcal{A}$ la courbe (*astroïde*) définie par les équations paramétriques :

$$\begin{cases} x(t) = \cos^3(t) \\ y(t) = \sin^3(t) \end{cases} \quad \text{pour } t \in [0, 2\pi].$$

1. Définir $x(t)$, $y(t)$ et tracer $\mathcal{A}$.
2. Calculer la longueur de $\mathcal{A}$ ainsi que l'aire qu'elle délimite.